Access-Version und Service Pack ermitteln
Versionen: Access 2003, 2002/XP, 2000 und 97

 FRAGE:  Mit "SysCmd(acSysCmdAccessVer)" kann ich feststellen, welche Access-Version eingesetzt wird. Wie aber bekomme ich heraus, welches Service Pack der Anwender installiert hat? Ich möchte bei fehlendem Service Pack gerne eine Warnmeldung ausgeben.
L. Brodersen

 ANTWORT:  Die folgende Funktion nutzt einen undokumentierten Parameter von "SysCmd", um die aktuelle Build-Nummer abzufragen. Daran kann man dann erkennen, welches Service Pack installiert ist:

Function GetAccVersAndSP() As String
  Dim strVers As String, strSP As String

  Select Case Val(SysCmd(acSysCmdAccessVer))
    Case 9 'Access 2000
      strVers = "Access 2000"
      Select Case SysCmd(715)
        Case 2719: strSP = "Kein SP!"
        Case Is >= 6620: strSP = "SP-3"
        Case Is >= 4506: strSP = "SP-2"
        Case Is >= 3822: strSP = "SP-1"
        Case Else: strSP = "Unbekanntes SP!"
      End Select

    Case 10 'Access 2002/XP
      strVers = "Access 2002/XP"
      Select Case SysCmd(715)
        Case 2627: strSP = "Kein SP!"
        Case Is >= 6501: strSP = "SP-3"
        Case Is >= 4302: strSP = "SP-2"
        Case Is >= 3409: strSP = "SP-1"
        Case Else: strSP = "Unbekanntes SP!"
      End Select

    Case 11 'Access 2003
      strVers = "Access 2003"
      Select Case SysCmd(715)
        Case 5614: strSP = "Kein SP!"
        Case Is >= 6566: strSP = "SP-2"
        Case Is >= 6355: strSP = "SP-1"
        Case Else: strSP = "Unbekanntes SP!"
      End Select
  End Select

  GetAccVersAndSP = strVers & ", " & strSP

End Function

